

Ideals, Varieties and Algorithms

David Cox, John Little, Donal O'Shea

Appendix C

Computer Algebra Systems

§2. Maple (updated March 3, 2010)

Our discussion applies to Maple 13. For us, the most important part of Maple is the *Groebner* package, though there is also the *PolynomialIdeals* package that will be discussed later in the section.

To have access to the commands in the *Groebner* package, type:

```
> with(Groebner)
```

(here, > is the Maple prompt). Once the *Groebner* package is loaded, you can perform the division algorithm, compute Groebner bases, and carry out a variety of other commands described below.

In Maple, a monomial ordering is called a monomial order. The monomial orderings lex, grlex, and grevlex from Chapter 2 are easy to use in Maple. Lex order is called `plex` (for “pure lexicographic”), grlex order is called `grlex`, and grevlex order is called `tdeg` (for “total degree”). Be careful not to confuse `tdeg` with `grlex`. Since a monomial order depends also on how the variables are ordered, Maple needs to know both the monomial order you want (`plex`, `grlex` or `tdeg`) and a list of variables. For example, to tell Maple to use lex order with variables $x > y > z$, you would need to input `plex(x,y,z)`.

The *Groebner* package also knows some elimination orders, as defined in Exercise 5 of Chapter 3, §1. To eliminate the first k variables from x_1, \dots, x_n , one can use the monomial order `lexdeg([x_1, ..., x_k], [x_{k+1}, ..., x_n])` (remember that Maple encloses a list inside brackets [...]). This order is the elimination order of Bayer and Stillman described in Exercise 6 of Chapter 3, §1.

The Maple documentation for the *Groebner* package also describes how to use certain weighted orders, and we will explain below how matrix orders give us many more monomial orderings.

The most commonly used commands in the *Groebner* package are `NormalForm`, for doing the division algorithm, and `Basis`, for computing a Groebner basis. `NormalForm` has the following syntax:

```
> NormalForm(f,polylist,monomialorder)
```

The output is the remainder of `f` on division by the polynomials in the list `polylist` using the monomial ordering specified by `monomialorder`. For example, to divide $x^3 + 3y^2$ by $x^2 + y$ and $x + 2xy$ using grevlex order with $x > y$, one would enter:

```
> NormalForm(x^3+3*y^2, [x^2+y,x+2*x*y], tdeg(x,y))
```

The base field here is the rational numbers \mathbb{Q} . To get the quotients in the division algorithm, add an optional letter, say \mathbb{Q} , as the fourth argument to `NormalForm`:

```
> NormalForm(f, polylist, monomialorder, Q)
```

This outputs the remainder as before, but if you ask Maple the value of \mathbb{Q} , it will return the list of quotients in the division algorithm.

As you might expect, `Basis` computes a Groebner basis, and the syntax is as follows:

```
> Basis(polylist, monomialorder)
```

This computes a Groebner basis for the ideal generated by the polynomials in `polylist` with respect to the monomial ordering specified by `monomialorder`. The answer is a reduced Groebner basis (in the sense of Chapter 2, §7), except for clearing denominators. As an example of how `gbasis` works, consider the command:

```
> gb := Basis([x^2+y, 2*x*y+y^2], plex(x, y))
```

This computes a list (and gives it the symbolic name `gb`) which is a Groebner basis for the ideal $\langle x^2 + y, 2xy + y^2 \rangle \subset \mathbb{Q}[x, y]$ using lex order with $x > y$. If you want to express the Groebner basis in terms of the polynomials in `polylist`, use `Basis` with the following optional argument:

```
> Basis(polylist, monomialorder, output=extended)
```

The output will be a pair (\mathbf{G}, \mathbf{C}) where \mathbf{G} is the Groebner basis and \mathbf{C} is a matrix of polynomials that expresses \mathbf{G} in terms of `polylist`.

If you use polynomials with integer or rational coefficients in `NormalForm` or `Basis`, Maple will assume that you are working over the field \mathbb{Q} . Note that there is no limitation on the size of the coefficients. Another possible coefficient field is the field of Gaussian rational numbers $\mathbb{Q}(i) = \{a + bi : a, b \in \mathbb{Q}\}$, where $i = \sqrt{-1}$. Note that Maple uses `I` to denote $\sqrt{-1}$. To compute a Groebner basis over a finite field with p elements (where p is a prime number), you need to include the option `characteristic=p` in the `Basis` command. (This option also works in `NormalForm`.)

Maple can also work with coefficients that lie in rational function fields. To tell Maple that a certain variable is in the base field (a “parameter”), you simply omit it from the variable list in the input. Thus, the command:

```
> Basis([v*x^2+y, u*x*y+y^2], plex(x, y))
```

will compute a Groebner basis for $\langle vx^2 + y, uxy + y^2 \rangle \subset \mathbb{Q}(u, v)[x, y]$ for lex order with $x > y$. The answer is reduced up to clearing denominators (so the leading coefficients of the Groebner basis are polynomials in u and v). (This also works in `NormalForm`.)

The *Groebner* package can work with matrix orders, where a matrix is regarded as a list of lists. Suppose that $[\mathbf{u}_1, \dots, \mathbf{u}_n]$ is a matrix, where each $\mathbf{u}_i = [u_{i1}, \dots, u_{in}]$ is a vector in \mathbb{Q}^n , where the entries of \mathbf{u}_1 are positive. Then define $x^\alpha > x^\beta$ if

$$\mathbf{u}_1 \cdot \alpha > \mathbf{u}_1 \cdot \beta, \text{ or } \mathbf{u}_1 \cdot \alpha = \mathbf{u}_1 \cdot \beta \text{ and } \mathbf{u}_2 \cdot \alpha > \mathbf{u}_2 \cdot \beta, \text{ or } \dots,$$

with ties (if any) broken by reverse lex. Orders of this type are discussed (from a slightly different point of view) in the remarks following Exercise 12 of Chapter 2, §4.

To see how such an order can be entered into Maple, suppose that we assign weights 1, 2, 3 to the variables x, y, z , so that the monomial $x^a y^b z^c$ has weighted degree $a + 2b + 3c$.

Then consider the monomial ordering that first compares weighted degree and then breaks ties using lex order with $x > y > z$. This is implemented in Maple by first defining the matrix M to be

```
> M := [[1,2,3],[1,0,0],[0,1,0]]
```

Then computations with this monomial order can be done by using `'matrix'(M, [x,y,z])` as the `monomialorder` in `NormalForm` or `Basis`. It is also possible to give this monomial order a symbolic name using the `MonomialOrder` command, which requires the *Ore_algebra* package. The documentation for `MonomialOrder` explains how this is done.

Some other useful Maple commands in the *Groebner* package are:

- `LeadingCoefficient`, `LeadingMonomial`, and `LeadingTerm`, which compute $LC(f)$, $LM(f)$, and $LT(f)$ for a polynomial f .
- `SPolynomial`, which computes the S-polynomial $S(f, g)$ of two polynomials.
- `IsProper`, which uses the consistency algorithm from Chapter 4, §1 to determine if a system of polynomial equations has a solution over an algebraically closed field.
- `IsZeroDimensional`, which uses the finiteness algorithm from Chapter 5, §3 to determine if a system of polynomial equations has finitely many solutions over an algebraically closed field.
- `UnivariatePolynomial`, which given a variable and a list of polynomials, computes the polynomial of lowest degree in the given variable which lies in the ideal generated by the polynomials.
- `HilbertPolynomial`, which given a list of polynomial, computes ${}^aHP_I(s) - {}^aHP_I(s-1)$ in the notation of Chapter 9, §3, where I is the ideal generated by the polynomials. When I is homogenous, Theorem 12 of Chapter 9, §3 shows that `HilbertPolynomial` computes the Hilbert polynomial $HP_I(s)$. A related command is `HilbertSeries`, which for a homogeneous ideal computes the Hilbert series as defined in Exercise 24 of Chapter 6, §4 of COX, LITTLE, and O'SHEA (1998).

There is also a `Solve` command which attempts to find all solutions of a system of equations. Maple has an excellent on-line help system that should make it easy to master these (and other) Maple commands in the *Groebner* package.

As already mentioned, Maple has the *PolynomialIdeals* package, which manipulates ideals in a polynomial ring. To load the package, type

```
> with(PolynomialIdeals)
```

Then an ideal such as $J = \langle x^2 + y, x - y^2 \rangle \subseteq \mathbb{Q}[x, y]$ is entered by typing

```
> J := <x^2+y,x-y^2>
```

Since Maple uses I to denote $\sqrt{-1}$, an ideal can't be named I unless you issue a command such as

```
> interface(imaginaryunit=_i)
```

This makes Maple use `_i` for $\sqrt{-1}$ and makes I available for the name of an ideal.

Some useful Maple commands in the *PolynomialIdeals* package are:

- `IdealMembership` tests ideal membership as discussed in Chapter 2, §8.
- `EliminationIdeal` computes the elimination ideals discussed in Chapter 3, §1.

- `RadicalMembership` tests radical membership as discussed in Chapter 4, §2. Related commands are `IsRadical` (test if an ideal is radical) and `Radical` (compute the radical of an ideal).
- `Intersect` computes the intersection of two ideals discussed in Chapter 4, §3.
- `Quotient` computes the ideal quotient $I : J$ discussed in Chapter 4, §4. The related command `Saturate` computes the saturation $I : J^\infty$.
- `PrimaryDecomposition` computes the primary decomposition of an ideal as discussed in Chapter 4, §7. A related command is `PrimeDecomposition`, which computes the prime decomposition of the radical of an ideal. This can be useful for solving systems of equations.

The *PolynomialIdeals* package contains many other commands that are useful dealing with polynomial ideals.

Finally, we should mention the existence of a Maple package written by Albert Lin and Philippe Loustau of George Mason University (with subsequent modifications by David Cox and Will Gryc of Amherst College and Chris Wensley of the University of Bangor, Wales) which complements the *Groebner* package. In this package, the program `div_alg` gives the quotients in the division algorithm, and the program `mxgb` computes a Groebner basis together with a matrix telling how to express the Groebner basis in terms of the given polynomials. This package is slow compared to the *Groebner* package, but can be used for many of the simpler examples in the book. There is also a Maple worksheet which explains how to use the package. Copies of the package and worksheet can be obtained from <http://www.cs.amherst.edu/~dac/iva.html>.